Find all spectroscopic terms for equivalent electrons-- D.English 3/5/05

Make a list of all $\{m_S, m_L\}$ for a given S and L

```
buildList[{S_, L_}] :=
 Flatten[
  Table[{ms, ml}, {ml, -L, L}, {ms, -S, S}],
  1]
```

Removes from a list all $\{m_S, m_L\}$ pairs for the highest m_S in alist

```
RemoveTerm[l_] := Block[{S = -l[[1, 1]], L = -l[[1, 2]], rlist, llen = Length[l], newl = l},
  rlist = buildList[{S, L}];
  (newl = DeleteCases[newl, #, 1, 1]) & /@ rlist;
  Print["Removed ", llen - Length[newl],
   "=", (2 S + 1) (2 L + 1), " terms associated with ", {S, L},
   DisplayForm[SuperscriptBox[" ", 2 S + 1]], StringTake["SPDFGHIJKLMNOP", {L + 1}]];
  newl
 ]
```

Recursively remove multiplets from alist until there are none left

```
IDall[l_] := If[Length[l] == 0, Print["All done"], IDall[RemoveTerm[l]]]
```

Make a list of all combinations of n elements of a list *l*

```
Combs[l_, n_] := Block[{},
  If[Length[l] == n, Return[{l}]];
  If[n == 0, Return[{{}}]];
  Join[
   Prepend[#, First[l]] & /@ Combs[Rest[l], n - 1],
   Combs[Rest[l], n]
  ]
 ]
```

Find all the terms for n electrons of type Lcode (a string

"s", "p", etc...)

```
Terms[n_, Lcode_] := Block[
  {L = StringPosition["spdfghijkl", Lcode][[1, 1]] - 1,
   termList, combList, totlList
  },
  termList = buildList[{1/2, L}];
  combList = Combs[termList, n];
  totlList = Total /@ combList // Sort;
  IDall[totlList];
 ]
```

Examples...

```
Terms[3, "p"]
```

Removed 4=4 terms associated with $\left\{\frac{3}{2}, 0\right\}$ ^{4}S

Removed 10=10 terms associated with $\left\{\frac{1}{2}, 2\right\}$ ^{2}D

Removed 6=6 terms associated with $\left\{\frac{1}{2}, 1\right\}$ ^{2}P

All done

```
Terms[2, "d"]
```

Removed 21=21 terms associated with {1, 3} ^{3}F

Removed 9=9 terms associated with {1, 1} ^{3}P

Removed 9=9 terms associated with {0, 4} ^{1}G

Removed 5=5 terms associated with {0, 2} ^{1}D

Removed 1=1 terms associated with {0, 0} ^{1}S

All done

```
Terms[3, "f"]
```

Removed 52=52 terms associated with $\left\{\frac{3}{2}, 6\right\}$ ^{4}I

Removed 36=36 terms associated with $\left\{\frac{3}{2}, 4\right\}$ ^{4}G

Removed 28=28 terms associated with $\left\{\frac{3}{2}, 3\right\}$ ^{4}F

Removed 20=20 terms associated with $\left\{\frac{3}{2}, 2\right\}$ ^{4}D

Removed 4=4 terms associated with $\left\{\frac{3}{2}, 0\right\}$ ^{4}S

Removed 34=34 terms associated with $\left\{\frac{1}{2}, 8\right\}$ ^{2}K

Removed 30=30 terms associated with $\left\{\frac{1}{2}, 7\right\}$ ^{2}J

Removed 26=26 terms associated with $\left\{\frac{1}{2}, 6\right\}$ ^{2}I

Removed 22=22 terms associated with $\left\{\frac{1}{2}, 5\right\}$ ^{2}H

Removed 22=22 terms associated with $\left\{\frac{1}{2}, 5\right\}$ ^{2}H

Removed 18=18 terms associated with $\left\{\frac{1}{2}, 4\right\}$ ^{2}G

Removed 18=18 terms associated with $\left\{\frac{1}{2}, 4\right\}$ ^{2}G

Removed 14=14 terms associated with $\left\{\frac{1}{2}, 3\right\}$ ^{2}F

Removed 14=14 terms associated with $\left\{\frac{1}{2}, 3\right\}$ ^{2}F

Removed 10=10 terms associated with $\left\{\frac{1}{2}, 2\right\}$ ^{2}D

Removed 10=10 terms associated with $\left\{\frac{1}{2}, 2\right\}$ ^{2}D

Removed 6=6 terms associated with $\left\{\frac{1}{2}, 1\right\}$ ^{2}P

All done